

```

+-----+
| MCU | ไทม์เคาน์เตอร์อินเตอร์รัพท์(Timer/Counter Interrupt)
| U06 | โดย อ.กำธร เรือนฝายกาฬ
+-----+

```

(1) กล่าวโดยทั่วไป

ภายใน AT90S2313 มีไทม์เคาน์เตอร์/เคาน์เตอร์ สองชุด ชุดที่หนึ่งเป็นแบบ 8 บิต ชุดที่สองเป็นแบบ 16 บิต

8-bit Timer/Counter0

=====

1. รีจิสเตอร์ TCCR0 - Timer/Counter0 Control Register

bit	7	6	5	4	3	2	1	0
	-	-	-	-	-	CS02	CS01	CS00
R/W	R	R	R	R	R	R/W	R/W	R/W
initial-value	0	0	0	0	0	0	0	0

CS02	CS01	CS00	Description
0	0	0	stop, the timer/counter 0
0	0	1	ck/1
0	1	0	ck/8
0	1	1	ck/64
1	0	0	ck/256
1	0	1	ck/1024
1	1	0	external (t0), falling edge
1	1	1	external (t0), rising edge

รีจิสเตอร์นี้ใช้กำหนดสัญญาณ clock ที่จะป้อนให้ตัวนับ(counter) เช่น ถ้าเราใช้คล็อก 4MHz และให้รีจิสเตอร์ TCCR0 มีค่าเท่ากับ 0b00000100 สัญญาณ clock 4MHz จะถูกหารด้วย 256 เท่ากับ 15625Hz แล้วนำไปเข้า เป็นสัญญาณ clock ของวงจรรนับ การนับของ timer/counter สแต็ปต่อสแต็ป เช่นนับจาก 00000000 เป็น 00000001 จะห่างกัน 64us เป็นต้น

2. รีจิสเตอร์ TCNT0 - Timer/Counter0

bit	7	6	5	4	3	2	1	0
	MSB							LSB
R/W	R/W	R/W	R/W	R/W	R/W	R/W	R/W	R/W
initial-value	0	0	0	0	0	0	0	0

รีจิสเตอร์นี้จะแสดงค่าที่กำลังนับอยู่ในปัจจุบันของ timer/counter เราสามารถอ่านหรือเขียนรีจิสเตอร์นี้ได้ การเขียนค่าลงไปนรีจิสเตอร์ตัวนี้ จะทำให้ค่า timer/counter ที่กำลังนับ กลายเป็นค่าใหม่ที่เราได้เขียนลงไป

3. Overflow interrupt หมายเลข 6 จาก Timer/Counter0 จะเกิดขึ้นทุกครั้งที่ counter เกิดการ overflow การที่จะให้อินเตอร์รัพท์นี้ทำงาน มีวิธีการดังนี้คือ

- 1) ตั้งอินเตอร์รัพท์เวกเตอร์ (interrupt vector) หมายเลข 6 ให้อยู่ที่อินเตอร์รัพท์รูทีนที่ต้องการให้ทำงาน ในที่นี้คือคำสั่ง

```
rcall I_7Segment ;018 #6 TIMERO OVFO
```

เป็นการตั้งให้อินเตอร์รัพท์นี้ กระโดดไปทำงานที่ I_7Segment เมื่อทำงานเสร็จก็ทำของอินเตอร์รัพท์รูทีนนี้ จะจบด้วยการสั่ง enable ให้อินเตอร์รัพท์นี้ทำงานอีกครั้งด้วยคำสั่ง sei (global interrupt enable) ตามด้วยคำสั่ง reti ซึ่งแทนที่จะเป็นคำสั่ง ret (subroutine return) เหมือนที่ผ่านมา

```
sei ;232 global interrupt enable
reti ;233 interrupt return
```

คำสั่ง reti ก็คือ interrupt return ซึ่งจะทำงานต่างกับคำสั่ง ret ก็คือคำสั่ง ret จะ pop ค่าของ PC คืนจากสแตค แต่คำสั่ง reti นอกจากจะ pop ค่า PC คืนจากสแตคแล้ว ยังยังเคลียร์บิต I ในรีจิสเตอร์ SREG ซึ่งเป็นบิตควบคุมการอินเตอร์รัพท์รวมของทุกอินเตอร์รัพท์

- 2) ทำการ enable ให้อินเตอร์รัพท์ timer0 overflow interrupt ทำงานโดยให้บิตที่ 1 คือบิต TOIE0 เป็นลอจิก 1

```
ldi r16, 0b00000010 ;303 enable TIMERO overflow int(#6)
out TMSK, r16 ;304 -
```

- 3) ตั้งบิต I ของ SREG เท่ากับ 1 ด้วยคำสั่ง sei(global interrupt enable)

```
sei ;305 global interrupt enable
```

- 4) ตั้งค่าตัวหารในรีจิสเตอร์ TCCR0 ตามที่เราต้องการ ในที่นี้ตั้งไว้เท่ากับ 256 ดังนั้นความถี่ในการเกิดอินเตอร์รัพท์จะเท่ากับ

$$4\text{MHz}/256/256 = 61\text{Hz}$$

ถ้าตั้งให้ตัวหารเท่ากับ 1024 คือให้ค่ารีจิสเตอร์ TCCR0 มีค่าเท่ากับ 00000101 จะได้ความถี่ในการเกิดอินเตอร์รัพท์หมายเลข 6 นี้เท่ากับ

$$4\text{MHz}/1024/256 = 15.3\text{Hz}$$

เลข 256 ตัวหลัง เป็นจำนวนสเต็ปที่ counter นับไปจนครบ 1 รอบ กล่าวคือ counter มีขนาด 8 บิต ดังนั้นจึงนับได้ตั้งแต่ 00000000, 00000001, 00000010, ... ไปจนถึง 11111111 รวมแล้วได้ 256 สเต็ป ก็จะเกิด timer0 overflow interrupt ขึ้นครั้งหนึ่งนั่นเอง

```
ldi r16, 0b00000100 ;306 TIMERO source = internal_clock/256
out TCCR0, r16 ;307 -
```

- 5) จากจุดนี้เป็นต้นไป Timer0 overflow interrupt ก็จะเริ่มทำงานทันที ถ้าได้โปรแกรมดูจะพบว่าในฟังก์ชัน main นั้นเราได้สั่งให้ MCU วนทำงานอยู่ในลูปไม่รู้จบคือ


```

;009 =====
;010 interrupt vector table
;011 =====
rjmp    F_main          ;012 #0 RESET
reti    ;013 #1 INTO
reti    ;014 #2 INT1
reti    ;015 #3 TIMER1 CAPT11
reti    ;016 #4 TIMER1 COMP1
reti    ;017 #5 TIMER1 OVFL
rcall   I_7Segment     ;018 #6 TIMER0 OVFO
reti    ;019 #7 UART,RX
reti    ;020 #8 UART,UDRE
reti    ;021 #9 UART,TX
reti    ;022 #10 ANA_COMP

;=====
;Function OutSegment
;{rin} r16=number to display 0..9
;      r17=digit select,0b111101111=3 0b111110111=2 0b111111011=0
;{rout} none
;{rdel} r0,r16,r17,Z
;{port} PORTD[x6543210] (segment)
;
;      ||||| |
;      ||||| | *** use with common cathode 7 segment only ***
;
;      ||||| |
;      ||||| | +- PD0 -----a-| - | -a-| - | -a-| - |
;      ||||| | +- PD1 -----b-| - | -b-| - | -b-| - |
;      ||||| | +- PD2 -----c-| | | -c-| | | -c-| | |
;      ||||| | +- PD3 -----d-| - | -d-| - | -d-| - |
;      ||||| | +- PD4 -----e-| | | -e-| | | -e-| | |
;      ||||| | +- PD5 -----f-| - * -f-| - * -f-| - *
;      ||||| | +- PD6 -----g-| - | -g-| - | -g-| - |
;
;      PORTB[x76543210] (digits)
;      |||
;      ||+--- digit 0 -----|-----|-----+
;      ||+--- digit 1 -----|-----+
;      ||+--- digit 3 -----+
;
;=====
F_OutSeg:
push    r0              ;100
push    r16             ;101
push    r17             ;102
push    ZL              ;103
push    ZH              ;104
ldi     ZL, LOW(LSegTab*2) ;105 get seg table base address
ldi     ZH, HIGH(LSegTab*2) ;106 -
cpi     r17, 0          ;107 if(digit=0)
brne   L_OutSegSkip1   ;108 else
ldi     r17, 0b11111011 ;109 then
rjmp   L_OutSegL1      ;110 -
L_OutSegSkip1:
cpi     r17, 1          ;111 if(digit=1)
brne   L_OutSegSkip2   ;112 else
ldi     r17, 0b11110111 ;113 then
rjmp   L_OutSegL1      ;114 -
L_OutSegSkip2:
cpi     r17, 2          ;115 if(digit=1)

```

```

        brne      L_OutSegExit      ;119 else
        ldi       r17, 0b11101111  ;120 then
        rjmp      L_OutSegL1       ;121 -
L_OutSegL1:
        cpi       r16,0            ;123 if(num==0)
        breq      L_OutSegL2       ;124 then jump to L_OutSegL2
        adiw      ZL, 2            ;125 else increase table offset
        subi      r16,1            ;126 - num=num-1
        rjmp      L_OutSegL1       ;127 - do it again
L_OutSegL2:
        lpm                          ;129 get program memory r0←(Z)
        ldi       r16, 0b11111111  ;130 port D : all pin=output
        out       DDRD, r16        ;131 -
        ldi       r16, 0b11111111  ;132 port B : all pin=output
        out       DDRB, r16        ;133 -
        out       PORTB, r17       ;134 select 7segment digit
        mov       r16, r0          ;135 output number to 7segment
        out       PORTD, r16       ;136 -
L_OutSegExit:
        ldi       r16, 5           ;138
        rcall     F_msDelay        ;139
        pop       ZH               ;140
        pop       ZL               ;141
        pop       r17              ;142
        pop       r16              ;143
        pop       r0               ;144
        ret                          ;145
;146 =====
;147 7segment lookup table
;148 =====
LSegTab:
        ;xabcdefg                ;149 number
        .DW      0b01111110      ;150 0          a
        .DW      0b00110000      ;151 1          ---
        .DW      0b01101101      ;152 2          f |   | b
        .DW      0b01111001      ;153 3          | g |
        .DW      0b00110011      ;154 4          ---
        .DW      0b01011011      ;155 5          e |   | c
        .DW      0b01011111      ;156 6          |   |
        .DW      0b01110000      ;157 7          ---
        .DW      0b01111111      ;158 8          d
        .DW      0b01111011      ;159 9

;.....

;=====
;Function 7Segment
;{rin} D_7Seg=number to display (3 digits ,range 0 to 127)
;{rout} none
;{rdel} none
;{port} none
;=====
I_7Segment:
        cli                          ;201 disable global interrupt
        push     r16                  ;202
        push     r17                  ;203
        push     r18                  ;204
        mov      r18, D_7Seg          ;205 store number
        ldi     r16, 0                ;206 reset counter
L_7Segment3cal:
        cpi     r18, 100              ;208 -

```

```

        brlt      L_7Segment3dis ;209 -
        inc      r16             ;210 -
        subi     r18, 100       ;211 -
        rjmp     L_7Segment3cal ;212 -
L_7Segment3dis:                ;213 -
        ldi     r17, 2         ;214 digit 2 selected
        rcall   F_OutSeg      ;215 -
        ldi     r16, 0         ;216 -
L_7Segment2cal:                ;217 digit 1 bin->decimal calculation
        cpi     r18, 10        ;218 -
        brlt   L_7Segment2dis ;219 -
        inc     r16            ;220 -
        subi   r18, 10        ;221 -
        rjmp   L_7Segment2cal ;222 -
L_7Segment2dis:                ;223 -
        ldi     r17, 1         ;224 digit 1 selected
        rcall   F_OutSeg      ;225 -
        mov     r16, r18       ;226 get number
        ldi     r17, 0         ;227 digit 0 selected
        rcall   F_OutSeg      ;228 -
        pop     r18            ;229
        pop     r17            ;230
        pop     r16            ;231
        sei     ;232 global interrupt enable
        reti    ;233 interrupt return
;.....

;=====
;Function msDelay for MCU working at 4MHz
;{rin} r16=delay time (unit in millisecc.) 0->255ms
;{rout} none
;{rdel} sreg,r16,r17,r18
;{port} none
;=====
F_msDelay:                       ;
        push   r16             ;
        push   r17             ;
        push   r18             ;
        cpi    r16, 0          ;
        breq   L_msDelayExit   ;
L_msDelayL3:                       ;
        ldi    r17, 11         ; 11
L_msDelayL2:                       ;
        ldi    r18, 113        ; 113
L_msDelayL1:                       ;
        dec    r18             ;
        brne   L_msDelayL1     ;
        dec    r17             ;
        brne   L_msDelayL2     ;
        dec    r16             ;
        brne   L_msDelayL3     ;
L_msDelayExit:                     ;
        pop    r18             ;
        pop    r17             ;
        pop    r16             ;
        ret     ;
;.....

```

```

;=====
;void main(void)
;=====
F_Main:
    ldi    r16,    low(RAMEND)    ;300 main function start here
    out    SPL,    r16            ;301 set stack pointer
    ldi    r16,    0b00000010    ;302 -
    out    TMSK,   r16            ;303 enable TIMER0 overflow interrupt(6)
    sei                                ;304 -
    ldi    r16,    0b00000100    ;305 global interrupt enable
    out    TCCR0,  r16            ;306 TIMER0 source = internal_clock/256
    ldi    r16,    0b00000101    ;307 -
    out    TCCR1B, r16            ;308 TIMER1 source = internal_clock/1024
    ldi    D_7Seg, 123            ;309 -
    ldi    D_7Seg, 123            ;310 7segment start value
L_MainLoop:
    rjmp   L_MainLoop            ;311
;.....

```